Pensieve header: A first implementation of nilpotent integration (first to use Picard iteration).

Initialization

```
In[∘]:= SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank"];
Once[<< KnotTheory`; << Rot.m];
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.

Loading Rot.m from http://drorbn.net/icbs24 to compute rotation numbers.
```

```
In[∘]:= CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
CCF[ℰ_] := Factor[ℰ];
CF[ω_. ℰ_𝔼] := CF[ω] × CF /@ ℰ;
CF[ℰ_List] := CF /@ ℰ;
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := Module[{vs = Cases[ℰ, (x | p)__, ∞] ⋃ {x, p, ϵ}, ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :> CCF[c] (Times @@ vs^ps)] ];
```

Integration

Using Picard Iteration!

```
In[∘]:= 𝔼 /: 𝔼[A_] × 𝔼[B_] := 𝔼[A + B]
```

```
In[∘]:= $π = Identity;
```

In[∘]:=
```
Unprotect[Integrate];
Integrate::sing = "How dare you ask me to integrate a singular Gaussian!";
∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, L0, Q, Δ, G, Z, e, λ, a, b},
   n = Length@vs; L0 = L /. ϵ → 0;
   Q = Table[(-∂_{vs⟦a⟧,vs⟦b⟧} L0) /. Thread[vs → 0] /. (p | x)__ → 0, {a, n}, {b, n}];
   If[(Δ = CF@Det[Q]) == 0, Message[Integrate::sing]; Return[]];
   Z = CF@$π[L + vs.Q.vs / 2]; G = Inverse[Q];
   While[
    e = CF@$π[(∂_λ Z) - 1/2 Sum_{a=1}^{n} Sum_{b=1}^{n} G⟦a, b⟧ ((∂_{vs⟦a⟧,vs⟦b⟧} Z) + (∂_{vs⟦a⟧} Z) (∂_{vs⟦b⟧} Z))];
    0 =!= e, Z -= ∫_0^λ e ⅆλ
   ];
   PowerExpand@Factor[ω (Δ (2 π)^n)^(-1/2)] × 𝔼[CF[Z /. λ → 1 /. Thread[vs → 0]]]
  ];
Protect[Integrate];
```

In[∘]:= $\int \mathbb{E}\left[i \lambda x_1^2/2\right] \mathbb{d}\{x_1\}$

Out[∘]=
$$\frac{(-1)^{1/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-i \lambda x_1^2/2\right] \mathbb{d}\{x_1\}$

Out[∘]=
$$-\frac{(-1)^{3/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[\frac{i}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\}\right] \mathbb{d}\{x_1, x_2\}$

Out[∘]=
$$\frac{\mathbb{E}[0]}{2\sqrt{b^2 - a c}\,\pi}$$

In[∘]:= $\int \mathbb{E}\left[-\lambda x_1^2/2\right] \mathbb{d}\{x_1\}$

Out[∘]=
$$\frac{\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-x_1^2/2 + \xi\, x_1\right] \mathrm{d}\{x_1\}$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{\xi^2}{2}\right]}{\sqrt{2\pi}}$$

In[∘]:= $\int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] \mathrm{d}\{x_1, x_2\}$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2 - 2\,b\,\xi_1\,\xi_2 + a\,\xi_2^2}{2\,(-b^2 + a\,c)}\right]}{2\sqrt{-b^2 + a\,c}\;\pi}$$

In[∘]:= $I1 = \int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] \mathrm{d}\{x_1\}$

Out[∘]=
$$\frac{\mathbb{E}\left[-\frac{(-b^2 + a\,c)\,x_2^2}{2\,a} + \frac{\xi_1^2}{2\,a} + \frac{x_2\,(-b\,\xi_1 + a\,\xi_2)}{a}\right]}{\sqrt{a}\;\sqrt{2\pi}}$$

In[∘]:= $\int I1\, \mathrm{d}\{x_2\}$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2 - 2\,b\,\xi_1\,\xi_2 + a\,\xi_2^2}{2\,(-b^2 + a\,c)}\right]}{2\sqrt{-b^2 + a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[\xi\, x + \eta\, y + z\,(x - y) + x^2\right] \mathrm{d}\{x, z\}$

Out[∘]=
$$-\frac{i\,\mathbb{E}[y\,(y + \eta + \xi)]}{2\pi}$$

Integration of ϵ-Series

In[∘]:= $\text{Block}\left[\left\{\$\pi = \text{Normal}\left[\# + O[\epsilon]^7\right]\&\right\},\ \int \mathbb{E}\left[-x^2/2 + \epsilon\, x^3/6\right] \mathrm{d}\{x\}\right]$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{5\,\epsilon^2}{24} + \frac{5\,\epsilon^4}{16} + \frac{1105\,\epsilon^6}{1152}\right]}{\sqrt{2\pi}}$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^7] &},
  ∫𝔼[-ϕ^2 / 2 + ϵ ϕ^4 / 24] ⅆ{ϕ}]
```

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{\epsilon}{8}+\frac{\epsilon^2}{12}+\frac{11\,\epsilon^3}{96}+\frac{17\,\epsilon^4}{72}+\frac{619\,\epsilon^5}{960}+\frac{709\,\epsilon^6}{324}\right]}{\sqrt{2\,\pi}}$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^5] &},
  ∫𝔼[p x + ϵ p^2 x] ⅆ{p, x}]
```

Out[∘]=

$$-\frac{i\,\mathbb{E}[0]}{2\,\pi}$$

In[∘]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
       mon ↦ And[
         Exponent[mon, ϵ] ≤ 2,
         Exponent[mon, x] == Exponent[mon, p]
        ]
      ] &},
  ∫𝔼[p x + a x^2 p + ϵ b x^3 p^3] ⅆ{p, x}]
```

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-6\,b\,\epsilon+342\,b^2\,\epsilon^2\right]}{2\,\pi}$$

In[∘]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
       mon ↦ And[
         Exponent[mon, ϵ] < 4,
         Exponent[mon, x] - Exponent[mon, p] ≤ 3
        ]
      ] &},
  ∫𝔼[p x + a x^2 p + ϵ b p^2 x] ⅆ{p, x}]
```

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-6\,a\,b\,\epsilon+162\,a^2\,b^2\,\epsilon^2-9072\,a^3\,b^3\,\epsilon^3\right]}{2\,\pi}$$

In[∘]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
      mon ↦ And[
        Exponent[mon, ϵ] < 4,
        Exponent[mon, x] - Exponent[mon, p] ≤ 3 - Exponent[mon, ϵ]
       ]
     ] &},
  ∫𝔼[p x + a x^2 p + ϵ b p^2 x] ⅆ{p, x}]
```

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-6\,a\,b\,\epsilon+162\,a^2\,b^2\,\epsilon^2-9072\,a^3\,b^3\,\epsilon^3\right]}{2\,\pi}$$

In[∘]:=
```
MatrixForm@Table[
  ∫𝔼[x_1 p_2 + x_2 p_3 + x_3 p_1 + ξ_i x_i + π_j p_j] ⅆ{x_1, x_2, x_3, p_1, p_2, p_3},
  {i, 3}, {j, 3}]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[-\pi_2\,\xi_1]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} \\ -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[-\pi_3\,\xi_2]}{8\,\pi^3} \\ -\frac{i\,\mathbb{E}[-\pi_1\,\xi_3]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} \end{pmatrix}$$

The ρ_1 Integrand

Adopted from pensieve://Projects/APAI/PerturbedGaussianIntegration.nb.

In[∘]:=
```
q[s_, i_, j_] := x_i ((1 - T^s) p_{i+1} + (T^s - 1) p_{j+1});
r_1[s_, i_, j_] :=
  s (-1 + 2 p_i x_i - 2 p_j x_i + (T^s - 1) p_i p_j x_i^2 + (1 - T^s) p_j^2 x_i^2 - 2 p_i p_j x_i x_j + 2 p_j^2 x_i x_j) / 2;
γ_1[φ_, k_] := ϵ φ (1 / 2 - x_k p_k);
ρ1i[s_, i_, j_] := T^(s/2) 𝔼[-q[s, i, j] + ϵ r_1[s, i, j]];
ρ1i[φ_, k_] := T^(φ/2) 𝔼[-x_k (p_k - p_{k+1}) + γ_1[φ, k]];
ρ1i[End, k_] := 𝔼[-x_k p_k];
ρ1i[K_] := Module[{Cs, φ, n, c, k, ℰ},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   ℰ = ρ1i[End, 2 n + 1];
   Do[ℰ *= ρ1i @@ c, {c, Cs}];
   Do[ℰ *= ρ1i[φ⟦k⟧, k], {k, 2 n}];
   CF@ℰ
  ];
ρ1vs[K_] := Union @@ Table[{x_i, p_i}, {i, 2 Crossings[K] + 1}]
```

In[∘]:=
```
ρ1i[Knot[3, 1]]
```

Out[∘]=

$$T^2\,\mathbb{E}\Big[-((p_1-p_2)\,x_1)-(p_2-p_3)\,x_2-\left(\left(1-\frac{1}{T}\right)p_3+\left(-1+\frac{1}{T}\right)p_6\right)x_2-(p_3-p_4)\,x_3-(p_4-p_5)\,x_4-\left(\left(-1+\frac{1}{T}\right)p_2+\left(1-\frac{1}{T}\right)p_5\right)x_4-\epsilon\left(\frac{1}{2}-p_4\,x_4\right)+\frac{1}{2}\epsilon\left(1+2\,p_1\,x_4-2\,p_4\,x_4-2\,p_1^2\,x_1\,x_4+2\,p_1\,p_4\,x_1\,x_4-\left(1-\frac{1}{T}\right)p_1^2\,x_4^2-\left(-1+\frac{1}{T}\right)p_1\,p_4\,x_4^2\right)-(p_5-p_6)\,x_5+\frac{1}{2}\epsilon\left(1-2\,p_2\,x_2+2\,p_5\,x_2-\left(-1+\frac{1}{T}\right)p_2\,p_5\,x_2^2-\left(1-\frac{1}{T}\right)p_5^2\,x_2^2+2\,p_2\,p_5\,x_2\,x_5-2\,p_5^2\,x_2\,x_5\right)-(p_6-p_7)\,x_6-\left(\left(-1+\frac{1}{T}\right)p_4+\left(1-\frac{1}{T}\right)p_7\right)x_6+\frac{1}{2}\epsilon\left(1+2\,p_3\,x_6-2\,p_6\,x_6-2\,p_3^2\,x_3\,x_6+2\,p_3\,p_6\,x_3\,x_6-\left(1-\frac{1}{T}\right)p_3^2\,x_6^2-\left(-1+\frac{1}{T}\right)p_3\,p_6\,x_6^2\right)-p_7\,x_7\Big]$$

In[∘]:=
```
ρ1vs[Knot[3, 1]]
```

Out[∘]=

$$\{p_1, p_2, p_3, p_4, p_5, p_6, p_7, x_1, x_2, x_3, x_4, x_5, x_6, x_7\}$$

In[∘]:=
```
K = Knot[5, 2];
Block[{$π = Normal[# + O[ϵ]^2] &},
 ρ1i[K]]
```

Out[∘]=

$$T^3\,\mathbb{E}\Big[-((p_1-p_2)\,x_1)-(p_2-p_3)\,x_2-\left(\left(1-\frac{1}{T}\right)p_3+\left(-1+\frac{1}{T}\right)p_8\right)x_2-(p_3-p_4)\,x_3-(p_4-p_5)\,x_4-\left(\left(-1+\frac{1}{T}\right)p_2+\left(1-\frac{1}{T}\right)p_5\right)x_4-\epsilon\left(\frac{1}{2}-p_4\,x_4\right)+\frac{1}{2}\epsilon\left(1+2\,p_1\,x_4-2\,p_4\,x_4-2\,p_1^2\,x_1\,x_4+2\,p_1\,p_4\,x_1\,x_4-\left(1-\frac{1}{T}\right)p_1^2\,x_4^2-\left(-1+\frac{1}{T}\right)p_1\,p_4\,x_4^2\right)-(p_5-p_6)\,x_5-(p_6-p_7)\,x_6-\left(\left(1-\frac{1}{T}\right)p_7+\left(-1+\frac{1}{T}\right)p_{10}\right)x_6-(p_7-p_8)\,x_7+\frac{1}{2}\epsilon\left(1-2\,p_2\,x_2+2\,p_7\,x_2-\left(-1+\frac{1}{T}\right)p_2\,p_7\,x_2^2-\left(1-\frac{1}{T}\right)p_7^2\,x_2^2+2\,p_2\,p_7\,x_2\,x_7-2\,p_7^2\,x_2\,x_7\right)-(p_8-p_9)\,x_8-\left(\left(-1+\frac{1}{T}\right)p_4+\left(1-\frac{1}{T}\right)p_9\right)x_8+\frac{1}{2}\epsilon\left(1+2\,p_3\,x_8-2\,p_8\,x_8-2\,p_3^2\,x_3\,x_8+2\,p_3\,p_8\,x_3\,x_8-\left(1-\frac{1}{T}\right)p_3^2\,x_8^2-\left(-1+\frac{1}{T}\right)p_3\,p_8\,x_8^2\right)-(p_9-p_{10})\,x_9+\epsilon\left(\frac{1}{2}-p_9\,x_9\right)+\frac{1}{2}\epsilon\left(1-2\,p_6\,x_6+2\,p_9\,x_6-\left(-1+\frac{1}{T}\right)p_6\,p_9\,x_6^2-\left(1-\frac{1}{T}\right)p_9^2\,x_6^2+2\,p_6\,p_9\,x_6\,x_9-2\,p_9^2\,x_6\,x_9\right)-(p_{10}-p_{11})\,x_{10}-\left(\left(-1+\frac{1}{T}\right)p_6+\left(1-\frac{1}{T}\right)p_{11}\right)x_{10}-\epsilon\left(\frac{1}{2}-p_{10}\,x_{10}\right)+\frac{1}{2}\epsilon\left(1+2\,p_5\,x_{10}-2\,p_{10}\,x_{10}-2\,p_5^2\,x_5\,x_{10}+2\,p_5\,p_{10}\,x_5\,x_{10}-\left(1-\frac{1}{T}\right)p_5^2\,x_{10}^2-\left(-1+\frac{1}{T}\right)p_5\,p_{10}\,x_{10}^2\right)-p_{11}\,x_{11}\Big]$$

In[∘]:=
```
K = Knot[5, 2];
Block[{$π = Normal[# + O[ϵ]^2] &},
 ∫ρ1i[K] ⅆ (ρ1vs@K)]
```

Out[∘]=

$$-\frac{i\, T^7\, \mathbb{E}\left[\frac{(-1+T)^2\,(5-4T+5T^2)\,\epsilon}{(2-3T+2T^2)^2}\right]}{2048\,\pi^{11}\,(2-3T+2T^2)}$$

In[∘]:=
```
K = Knot[8, 19];
Block[{$π = Normal[# + O[ϵ]^2] &},
 ∫ρ1i@K ⅆ (ρ1vs@K)]
```

Out[∘]=

$$-\frac{i\, \mathbb{E}\left[-\frac{(-1+T)^2\,(1+T^4)\,(3+4T^3+3T^6)\,\epsilon}{(1-T+T^2)^2\,(1-T^2+T^4)^2}\right]}{131072\,\pi^{17}\,T^5\,(1-T+T^3-T^5+T^6)}$$

Concatenating edges

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 lhs = ∫(𝔼[π_i p_i] × ρ1i[φ1, i] × ρ1i[φ2, i + 1]) ⅆ {x_i, p_i, x_{i+1}, p_{i+1}}]
```

Out[∘]=

$$\frac{T^{\frac{\varphi 1}{2}+\frac{\varphi 2}{2}}\, \mathbb{E}\left[\frac{1}{2}\epsilon\,(-\varphi 1-\varphi 2)+p_{2+i}\,\pi_i-\epsilon\,(\varphi 1+\varphi 2)\,p_{2+i}\,\pi_i\right]}{4\,\pi^2}$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 lhs = ∫(𝔼[π_i p_i] × ρ1i[φ1 + φ2, i]) ⅆ {x_i, p_i}]
```

Out[∘]=

$$-\frac{i\, T^{\frac{\varphi 1}{2}+\frac{\varphi 2}{2}}\, \mathbb{E}\left[\frac{1}{2}\epsilon\,(-\varphi 1-\varphi 2)+p_{1+i}\,\pi_i-\epsilon\,(\varphi 1+\varphi 2)\,p_{1+i}\,\pi_i\right]}{2\,\pi}$$

Invariance Under Reidemeister 3b

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 lhs =
  ∫(𝔼[π_i p_i + π_j p_j + π_k p_k] × ρ1i[1, i, j] × ρ1i[1, i + 1, k] × ρ1i[1, j + 1, k + 1] × ρ1i[0, i] ×
     ρ1i[0, j] × ρ1i[0, k] × ρ1i[0, i + 1] × ρ1i[0, j + 1] × ρ1i[0, k + 1])
   ⅆ{x_i, x_j, x_k, p_i, p_j, p_k, x_{i+1}, x_{j+1}, x_{k+1}, p_{i+1}, p_{j+1}, p_{k+1}}]
```

Out[∘]=

$$\frac{1}{64\pi^6} T^{3/2}\,\mathbb{E}\Big[-\frac{3\epsilon}{2}+T^2 p_{2+i}\pi_i+\frac{1}{2}T^3\epsilon\, p_{2+i}p_{2+j}\pi_i(-\pi_i+T\pi_i-2\pi_j)-\frac{1}{2}T^3\epsilon\, p_{2+j}^2\pi_i(-\pi_i+T\pi_i-2\pi_j)+T\epsilon\, p_{2+j}(T\pi_i-\pi_j)-T p_{2+j}(-\pi_i+T\pi_i-\pi_j)-\frac{1}{2}T\epsilon\, p_{2+k}^2(\pi_i+\pi_j)(-\pi_i+T\pi_i-\pi_j+T\pi_j-2\pi_k)+\frac{1}{2}T^2\epsilon\, p_{2+i}p_{2+k}\pi_i(-\pi_i+T\pi_i-2\pi_j+2T\pi_j-2\pi_k)+\epsilon\, p_{2+k}(T\pi_i-\pi_j+2T\pi_j-2\pi_k)+p_{2+k}(\pi_i-T\pi_i+\pi_j-T\pi_j+\pi_k)-\frac{1}{2}T\epsilon\, p_{2+j}p_{2+k}\left(\pi_i^2-2T\pi_i^2+T^2\pi_i^2+2\pi_i\pi_j-4T\pi_i\pi_j+2T^2\pi_i\pi_j+\pi_j^2-T\pi_j^2+2\pi_i\pi_k-2T\pi_i\pi_k+2\pi_j\pi_k\right)\Big]$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 rhs =
  ∫(𝔼[π_i p_i + π_j p_j + π_k p_k] × ρ1i[1, j, k] × ρ1i[1, i, k + 1] × ρ1i[1, i + 1, j + 1] × ρ1i[0, i] ×
     ρ1i[0, j] × ρ1i[0, k] × ρ1i[0, i + 1] × ρ1i[0, j + 1] × ρ1i[0, k + 1])
   ⅆ{x_i, x_j, x_k, p_i, p_j, p_k, x_{i+1}, x_{j+1}, x_{k+1}, p_{i+1}, p_{j+1}, p_{k+1}}]
```

Out[∘]=

$$\frac{1}{64\pi^6} T^{3/2}\,\mathbb{E}\Big[-\frac{3\epsilon}{2}+T^2 p_{2+i}\pi_i+\frac{1}{2}T^3\epsilon\, p_{2+i}p_{2+j}\pi_i(-\pi_i+T\pi_i-2\pi_j)-\frac{1}{2}T^3\epsilon\, p_{2+j}^2\pi_i(-\pi_i+T\pi_i-2\pi_j)+T\epsilon\, p_{2+j}(T\pi_i-\pi_j)-T p_{2+j}(-\pi_i+T\pi_i-\pi_j)-\frac{1}{2}T\epsilon\, p_{2+k}^2(\pi_i+\pi_j)(-\pi_i+T\pi_i-\pi_j+T\pi_j-2\pi_k)+\frac{1}{2}T^2\epsilon\, p_{2+i}p_{2+k}\pi_i(-\pi_i+T\pi_i-2\pi_j+2T\pi_j-2\pi_k)+\epsilon\, p_{2+k}(T\pi_i-\pi_j+2T\pi_j-2\pi_k)+p_{2+k}(\pi_i-T\pi_i+\pi_j-T\pi_j+\pi_k)-\frac{1}{2}T\epsilon\, p_{2+j}p_{2+k}\left(\pi_i^2-2T\pi_i^2+T^2\pi_i^2+2\pi_i\pi_j-4T\pi_i\pi_j+2T^2\pi_i\pi_j+\pi_j^2-T\pi_j^2+2\pi_i\pi_k-2T\pi_i\pi_k+2\pi_j\pi_k\right)\Big]$$

In[∘]:=
```
lhs == rhs
```

Out[∘]=
```
True
```

Invariance Under Reidemeister 2b

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
  lhs = ∫(𝔼[π_i p_i + π_j p_j + ξ_{i+1} x_{i+1} + ξ_{j+1} x_{j+1}] × ρ1i[1, i, j] × ρ1i[-1, i + 1, j + 1] × ρ1i[0, i] ×
      ρ1i[0, j] × ρ1i[0, i + 1] × ρ1i[0, j + 1]) ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}]
```

Out[∘]=

$$\frac{\mathbb{E}[p_{2+i}\,\pi_i + p_{2+j}\,\pi_j + T\,\pi_i\,\xi_{1+i} + \pi_i\,\xi_{1+j} - T\,\pi_i\,\xi_{1+j} + \pi_j\,\xi_{1+j}]}{16\,\pi^4}$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
  rhs =
   ∫(𝔼[π_i p_i + π_j p_j + ξ_{i+1} x_{i+1} + ξ_{j+1} x_{j+1}] × ρ1i[0, i] × ρ1i[0, j] × ρ1i[0, i + 1] × ρ1i[0, j + 1])
    ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}]
```

Out[∘]=

$$\frac{\mathbb{E}[p_{2+i}\,\pi_i + p_{2+j}\,\pi_j + \pi_i\,\xi_{1+i} + \pi_j\,\xi_{1+j}]}{16\,\pi^4}$$

In[∘]:=
```
lhs == rhs
```

Out[∘]=

$$\frac{\mathbb{E}[p_{2+i}\,\pi_i + p_{2+j}\,\pi_j + T\,\pi_i\,\xi_{1+i} + \pi_i\,\xi_{1+j} - T\,\pi_i\,\xi_{1+j} + \pi_j\,\xi_{1+j}]}{16\,\pi^4} == \frac{\mathbb{E}[p_{2+i}\,\pi_i + p_{2+j}\,\pi_j + \pi_i\,\xi_{1+i} + \pi_j\,\xi_{1+j}]}{16\,\pi^4}$$

Invariance Under R2c



In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
  lhs = ∫(𝔼[π_i p_i + π_j p_j] × ρ1i[-1, i, j + 1] × ρ1i[1, i + 1, j] × ρ1i[0, i] ×
      ρ1i[0, j] × ρ1i[0, i + 1] × ρ1i[1, j + 1]) ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}]
```

Out[∘]=

$$\frac{\sqrt{T}\,\mathbb{E}\left[-\frac{\epsilon}{2} + p_{2+i}\,\pi_i + p_{2+j}\,\pi_j - \epsilon\,p_{2+j}\,\pi_j\right]}{16\,\pi^4}$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 rhs = ∫(𝔼[π_i p_i + π_j p_j] × ρ1i[0, i] × ρ1i[0, j] × ρ1i[0, i + 1] × ρ1i[1, j + 1])
   ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}]
```

Out[∘]=

$$\frac{\sqrt{T}\ \mathbb{E}\left[-\frac{\epsilon}{2}+p_{2+i}\,\pi_i+p_{2+j}\,\pi_j-\epsilon\,p_{2+j}\,\pi_j\right]}{16\,\pi^4}$$

In[∘]:=
```
lhs == rhs
```

Out[∘]=

True

Invariance Under R1l



In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 lhs = ∫(𝔼[π_i p_i] × ρ1i[1, i + 1, i] × ρ1i[0, i] × ρ1i[1, i + 1]) ⅆ{x_i, p_i, x_{i+1}, p_{i+1}}]
```

Out[∘]=

$$\frac{\mathbb{E}[p_{2+i}\,\pi_i]}{4\,\pi^2}$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 rhs = ∫(𝔼[π_i p_i] × ρ1i[0, i] × ρ1i[0, i + 1]) ⅆ{x_i, p_i, x_{i+1}, p_{i+1}}]
```

Out[∘]=

$$\frac{\mathbb{E}[p_{2+i}\,\pi_i]}{4\,\pi^2}$$

In[∘]:=
```
lhs == rhs
```

Out[∘]=

True

Invariance Under R1r



In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 lhs = ∫ (𝔼[π_i p_i] × ρ1i[1, i, i + 1] × ρ1i[0, i] × ρ1i[-1, i + 1]) ⅆ{x_i, p_i, x_{i+1}, p_{i+1}}]
```

Out[∘]=

$$\frac{\mathbb{E}[p_{2+i}\,\pi_i]}{4\,\pi^2}$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 rhs = ∫ (𝔼[π_i p_i] × ρ1i[0, i] × ρ1i[0, i + 1]) ⅆ{x_i, p_i, x_{i+1}, p_{i+1}}]
```

Out[∘]=

$$\frac{\mathbb{E}[p_{2+i}\,\pi_i]}{4\,\pi^2}$$

In[∘]:=
```
lhs == rhs
```

Out[∘]=

True

Invariance Under Sw



In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 lhs = ∫ (𝔼[π_i p_i + π_j p_j] × ρ1i[1, i, j] × ρ1i[-1, i] × ρ1i[1, i + 1] × ρ1i[-1, j] × ρ1i[1, j + 1])
   ⅆ{x_i, x_j, p_i, p_j}]
```

Out[∘]=

$$\frac{1}{4\,\pi^2}\sqrt{T}\,\mathbb{E}\Big[\frac{3\,\epsilon}{2} + T\,p_{1+i}\,\pi_i + T\,\epsilon\,p_{1+i}\,\pi_i + \epsilon\,p_{1+j}\,\pi_i + \frac{1}{2}\,T\,\epsilon\,p_{1+i}\,p_{1+j}\,\pi_i\,(-\pi_i + T\,\pi_i - 2\,\pi_j) - \frac{1}{2}\,T\,\epsilon\,p_{1+j}^2\,\pi_i\,(-\pi_i + T\,\pi_i - 2\,\pi_j) + p_{1+j}\,(\pi_i - T\,\pi_i + \pi_j) - p_{1+i}\,x_{1+i} - \epsilon\,p_{1+i}\,x_{1+i} + p_{2+i}\,x_{1+i} - p_{1+j}\,x_{1+j} - \epsilon\,p_{1+j}\,x_{1+j} + p_{2+j}\,x_{1+j}\Big]$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 rhs = ∫(𝔼[π_i p_i + π_j p_j] × ρ1i[1, i, j] × ρ1i[0, i] × ρ1i[0, i + 1] × ρ1i[0, j] × ρ1i[0, j + 1])
   ⅆ{x_i, x_j, p_i, p_j}]
```

Out[∘]=

$$\frac{1}{4\pi^2}\sqrt{T}\,\mathbb{E}\Big[-\frac{\epsilon}{2}+T\,p_{1+i}\,\pi_i+\frac{1}{2}\,T\,\epsilon\,p_{1+i}\,p_{1+j}\,\pi_i\,(-\pi_i+T\,\pi_i-2\,\pi_j)-\frac{1}{2}\,T\,\epsilon\,p_{1+j}^2\,\pi_i\,(-\pi_i+T\,\pi_i-2\,\pi_j)+\epsilon\,p_{1+j}\,(T\,\pi_i-\pi_j)+p_{1+j}\,(\pi_i-T\,\pi_i+\pi_j)-p_{1+i}\,x_{1+i}+p_{2+i}\,x_{1+i}-p_{1+j}\,x_{1+j}+p_{2+j}\,x_{1+j}\Big]$$

In[∘]:=
```
lhs == rhs
```

Out[∘]=

$$\frac{1}{4\pi^2}\sqrt{T}\,\mathbb{E}\Big[\frac{3\,\epsilon}{2}+T\,p_{1+i}\,\pi_i+T\,\epsilon\,p_{1+i}\,\pi_i+\epsilon\,p_{1+j}\,\pi_i+\frac{1}{2}\,T\,\epsilon\,p_{1+i}\,p_{1+j}\,\pi_i\,(-\pi_i+T\,\pi_i-2\,\pi_j)-\frac{1}{2}\,T\,\epsilon\,p_{1+j}^2\,\pi_i\,(-\pi_i+T\,\pi_i-2\,\pi_j)+p_{1+j}\,(\pi_i-T\,\pi_i+\pi_j)-p_{1+i}\,x_{1+i}-\epsilon\,p_{1+i}\,x_{1+i}+p_{2+i}\,x_{1+i}-p_{1+j}\,x_{1+j}-\epsilon\,p_{1+j}\,x_{1+j}+p_{2+j}\,x_{1+j}\Big]==\frac{1}{4\pi^2}\sqrt{T}\,\mathbb{E}\Big[-\frac{\epsilon}{2}+T\,p_{1+i}\,\pi_i+\frac{1}{2}\,T\,\epsilon\,p_{1+i}\,p_{1+j}\,\pi_i\,(-\pi_i+T\,\pi_i-2\,\pi_j)-\frac{1}{2}\,T\,\epsilon\,p_{1+j}^2\,\pi_i\,(-\pi_i+T\,\pi_i-2\,\pi_j)+\epsilon\,p_{1+j}\,(T\,\pi_i-\pi_j)+p_{1+j}\,(\pi_i-T\,\pi_i+\pi_j)-p_{1+i}\,x_{1+i}+p_{2+i}\,x_{1+i}-p_{1+j}\,x_{1+j}+p_{2+j}\,x_{1+j}\Big]$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 lhs = ∫(𝔼[π_i p_i + π_j p_j + π_{i+1} p_{i+1} + π_{j+1} p_{j+1} + ξ_{i+1} x_{i+1} + ξ_{j+1} x_{j+1}] × ρ1i[1, i, j] × ρ1i[-1, i] ×
    ρ1i[1, i + 1] × ρ1i[-1, j] × ρ1i[1, j + 1]) ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}]
```

Out[∘]=

$$\frac{1}{16\pi^4}\sqrt{T}\,\mathbb{E}\Big[p_{2+i}\,(T\,\pi_i+\pi_{1+i})+\frac{1}{2}\,T\,\epsilon\,p_{2+i}\,p_{2+j}\,\pi_i\,(-\pi_i+T\,\pi_i-2\,\pi_j)-\frac{1}{2}\,T\,\epsilon\,p_{2+j}^2\,\pi_i\,(-\pi_i+T\,\pi_i-2\,\pi_j)+p_{2+j}\,(\pi_i-T\,\pi_i+\pi_j+\pi_{1+j})+T\,\pi_i\,\xi_{1+i}+\pi_{1+i}\,\xi_{1+i}+\pi_i\,\xi_{1+j}-T\,\pi_i\,\xi_{1+j}+\pi_j\,\xi_{1+j}+\pi_{1+j}\,\xi_{1+j}+\frac{1}{2}\,\epsilon\,p_{2+i}\,\left(-2\,\pi_{1+i}-T\,\pi_i^2\,\xi_{1+j}+T^2\,\pi_i^2\,\xi_{1+j}-2\,T\,\pi_i\,\pi_j\,\xi_{1+j}\right)+\frac{1}{2}\,\epsilon\,p_{2+j}\,\left(2\,T\,\pi_i-2\,\pi_j-2\,\pi_{1+j}-T\,\pi_i^2\,\xi_{1+i}+T^2\,\pi_i^2\,\xi_{1+i}-2\,T\,\pi_i\,\pi_j\,\xi_{1+i}+2\,T\,\pi_i^2\,\xi_{1+j}-2\,T^2\,\pi_i^2\,\xi_{1+j}+4\,T\,\pi_i\,\pi_j\,\xi_{1+j}\right)+\frac{1}{2}\,\epsilon\,\left(-1-2\,\pi_{1+i}\,\xi_{1+i}+2\,T\,\pi_i\,\xi_{1+j}-2\,\pi_j\,\xi_{1+j}-2\,\pi_{1+j}\,\xi_{1+j}-T\,\pi_i^2\,\xi_{1+i}\,\xi_{1+j}+T^2\,\pi_i^2\,\xi_{1+i}\,\xi_{1+j}-2\,T\,\pi_i\,\pi_j\,\xi_{1+i}\,\xi_{1+j}+T\,\pi_i^2\,\xi_{1+j}^2-T^2\,\pi_i^2\,\xi_{1+j}^2+2\,T\,\pi_i\,\pi_j\,\xi_{1+j}^2\right)\Big]$$

In[∘]:=
```
Block[{$π = Normal[# + O[ϵ]^2] &},
 rhs = ∫(𝔼[π_i p_i + π_j p_j + π_{i+1} p_{i+1} + π_{j+1} p_{j+1} + ξ_{i+1} x_{i+1} + ξ_{j+1} x_{j+1}] × ρ1i[1, i, j] × ρ1i[0, i] ×
     ρ1i[0, i + 1] × ρ1i[0, j] × ρ1i[0, j + 1]) ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}]
```

Out[∘]=

$$\frac{1}{16\pi^4}\sqrt{T}\,\mathbb{E}\Big[p_{2+i}\,(T\pi_i+\pi_{1+i})+\frac{1}{2}T\epsilon\,p_{2+i}\,p_{2+j}\,\pi_i\,(-\pi_i+T\pi_i-2\pi_j)-\frac{1}{2}T\epsilon\,p_{2+j}^2\,\pi_i\,(-\pi_i+T\pi_i-2\pi_j)+p_{2+j}\,(\pi_i-T\pi_i+\pi_j+\pi_{1+j})+T\pi_i\,\xi_{1+i}+\pi_{1+i}\,\xi_{1+i}+\pi_i\,\xi_{1+j}-T\pi_i\,\xi_{1+j}+\frac{1}{2}T\epsilon\,p_{2+i}\,\pi_i\,(-\pi_i+T\pi_i-2\pi_j)\,\xi_{1+j}+\pi_j\,\xi_{1+j}+\pi_{1+j}\,\xi_{1+j}+\frac{1}{2}\epsilon\,p_{2+j}\,\big(2T\pi_i-2\pi_j-T\pi_i^2\,\xi_{1+i}+T^2\pi_i^2\,\xi_{1+i}-2T\pi_i\,\pi_j\,\xi_{1+i}+2T\pi_i^2\,\xi_{1+j}-2T^2\pi_i^2\,\xi_{1+j}+4T\pi_i\,\pi_j\,\xi_{1+j}\big)+\frac{1}{2}\epsilon\,\big(-1+2T\pi_i\,\xi_{1+j}-2\pi_j\,\xi_{1+j}-T\pi_i^2\,\xi_{1+i}\,\xi_{1+j}+T^2\pi_i^2\,\xi_{1+i}\,\xi_{1+j}-2T\pi_i\,\pi_j\,\xi_{1+i}\,\xi_{1+j}+T\pi_i^2\,\xi_{1+j}^2-T^2\pi_i^2\,\xi_{1+j}^2+2T\pi_i\,\pi_j\,\xi_{1+j}^2\big)\Big]$$

In[∘]:=
```
lhs == rhs
```

Out[∘]=

$$\frac{1}{16\pi^4}\sqrt{T}\,\mathbb{E}\Big[p_{2+i}\,(T\pi_i+\pi_{1+i})+\frac{1}{2}T\epsilon\,p_{2+i}\,p_{2+j}\,\pi_i\,(-\pi_i+T\pi_i-2\pi_j)-\frac{1}{2}T\epsilon\,p_{2+j}^2\,\pi_i\,(-\pi_i+T\pi_i-2\pi_j)+p_{2+j}\,(\pi_i-T\pi_i+\pi_j+\pi_{1+j})+T\pi_i\,\xi_{1+i}+\pi_{1+i}\,\xi_{1+i}+\pi_i\,\xi_{1+j}-T\pi_i\,\xi_{1+j}+\pi_j\,\xi_{1+j}+\pi_{1+j}\,\xi_{1+j}+\frac{1}{2}\epsilon\,p_{2+i}\,\big(-2\pi_{1+i}-T\pi_i^2\,\xi_{1+j}+T^2\pi_i^2\,\xi_{1+j}-2T\pi_i\,\pi_j\,\xi_{1+j}\big)+\frac{1}{2}\epsilon\,p_{2+j}\,\big(2T\pi_i-2\pi_j-2\pi_{1+j}-T\pi_i^2\,\xi_{1+i}+T^2\pi_i^2\,\xi_{1+i}-2T\pi_i\,\pi_j\,\xi_{1+i}+2T\pi_i^2\,\xi_{1+j}-2T^2\pi_i^2\,\xi_{1+j}+4T\pi_i\,\pi_j\,\xi_{1+j}\big)+\frac{1}{2}\epsilon\,\big(-1-2\pi_{1+i}\,\xi_{1+i}+2T\pi_i\,\xi_{1+j}-2\pi_j\,\xi_{1+j}-2\pi_{1+j}\,\xi_{1+j}-T\pi_i^2\,\xi_{1+i}\,\xi_{1+j}+T^2\pi_i^2\,\xi_{1+i}\,\xi_{1+j}-2T\pi_i\,\pi_j\,\xi_{1+i}\,\xi_{1+j}+T\pi_i^2\,\xi_{1+j}^2-T^2\pi_i^2\,\xi_{1+j}^2+2T\pi_i\,\pi_j\,\xi_{1+j}^2\big)\Big]==$$

$$\frac{1}{16\pi^4}\sqrt{T}\,\mathbb{E}\Big[p_{2+i}\,(T\pi_i+\pi_{1+i})+\frac{1}{2}T\epsilon\,p_{2+i}\,p_{2+j}\,\pi_i\,(-\pi_i+T\pi_i-2\pi_j)-\frac{1}{2}T\epsilon\,p_{2+j}^2\,\pi_i\,(-\pi_i+T\pi_i-2\pi_j)+p_{2+j}\,(\pi_i-T\pi_i+\pi_j+\pi_{1+j})+T\pi_i\,\xi_{1+i}+\pi_{1+i}\,\xi_{1+i}+\pi_i\,\xi_{1+j}-T\pi_i\,\xi_{1+j}+\frac{1}{2}T\epsilon\,p_{2+i}\,\pi_i\,(-\pi_i+T\pi_i-2\pi_j)\,\xi_{1+j}+\pi_j\,\xi_{1+j}+\pi_{1+j}\,\xi_{1+j}+\frac{1}{2}\epsilon\,p_{2+j}\,\big(2T\pi_i-2\pi_j-T\pi_i^2\,\xi_{1+i}+T^2\pi_i^2\,\xi_{1+i}-2T\pi_i\,\pi_j\,\xi_{1+i}+2T\pi_i^2\,\xi_{1+j}-2T^2\pi_i^2\,\xi_{1+j}+4T\pi_i\,\pi_j\,\xi_{1+j}\big)+\frac{1}{2}\epsilon\,\big(-1+2T\pi_i\,\xi_{1+j}-2\pi_j\,\xi_{1+j}-T\pi_i^2\,\xi_{1+i}\,\xi_{1+j}+T^2\pi_i^2\,\xi_{1+i}\,\xi_{1+j}-2T\pi_i\,\pi_j\,\xi_{1+i}\,\xi_{1+j}+T\pi_i^2\,\xi_{1+j}^2-T^2\pi_i^2\,\xi_{1+j}^2+2T\pi_i\,\pi_j\,\xi_{1+j}^2\big)\Big]$$